

第 3 章

表計算ソフトウェア：微積分，行列の概念の学修

2024 年 11 月 18 日

学習目標

- (1) 表計算ソフトウェアの表を通して微分の概念を理解し，確認する (関数の変化率).
- (2) 表計算ソフトウェアの表を通して積分の概念を理解する (累積値，面積).
- (3) 金利計算を通して指数・対数関数を理解する.
- (4) 表計算ソフトウェアを通して，行列の概念を理解する.

本章は，専修大学商学部の高萩栄一郎の著作である．

1 微分の概念

1.1 微分の意味 1

本節では、微分とはなにか、何を表すのかを理解し、その意味を表計算ソフトウェアの式の設定とその計算結果により確認します。

微分は、関数 $y = f(x)$ を考えたとき、ある x で x をほんの少し増加 (Δx) させたとき、 y の値の増加量 (Δy) の比率 $\Delta y / \Delta x$ (変化率) を求めることを言います。(もう少し正確にいうと、 Δx を 0 に近づけていった極限の値) もう少し、ざっくりとした表現でいうと、「ある x で、 x をちょっと増やしたら、 y はどれくらい増減するのか (変化率)」です。

元の関数から、さまざまな x で、この y の値の増加量 (Δy) と x の増分 (Δx) の比率 $\Delta y / \Delta x$ の関数を導関数と呼びます。関数 $f(x)$ の導関数は、 $f'(x)$ と記述します。導関数は、変化率の関数とも言われています。

たとえば、 x を所得額、 y を所得税額としたとき、所得が 100 円増えたとき、所得税額はどれくらい増えるのかにあたります。例えば、所得税額が 20 円増えたら、 $\Delta y / \Delta x = 0.2 = 20\%$ で、これは限界税率と呼ばれています。ある x で限界税率が 100% を越えたら、所得の増加分より大きな所得税が増加するわけですから、多くの人は、その x で所得を増やそうとは思わないでしょう。

$f(x)$ を利益関数として、 x を生産量とします。いろいろな生産量 x で、 $f(x)$ を求め、その x での $\Delta y / \Delta x$ を求めます (この関数は、 $f(x)$ の導関数 $f'(x)$ で計算できます)。ある x で、 $\Delta y / \Delta x$ が正であれば、生産量 x を増やそうと考えます。 x を増やしていくうちに、 $\Delta y / \Delta x$ がだんだん小さくなり、やがて負になったとします。負であれば、利益が減るので、その時点で x を増やさないでしょう。ちょうど $\Delta y / \Delta x$ が 0 のとき、利益が最大となるので、このときの x に生産量を決めるということが想定されます。これが、「微分して 0 と置く」という x の計算方法です。

このように微分の考え方は、人文社会科学でもいろいろなところで利用されています。式の微分を計算できるということも重要ですが、著者は、微分という「ことば」がどういう意味を持っているのか、どういう性質を持っているのかを理解す

ることがまず重要だと思います。式を微分して導関数を求めたりすることは、数式処理ソフトウェアを使えば、多くの式の場合、簡単に求めることができます。このテキストでも Web 上の微分・積分のサイトを利用します（後述）。そこで、微分の考え方を表計算ソフトウェア上で表現して、その計算結果が、微分・積分のサイトの式の計算結果にほぼ一致するのか確認し、理解を深めたいと思います。

1.2 微分の意味 2

微分の値は、 $\Delta y / \Delta x$ （変化率）について、 Δx を 0 に近づけていった極限の値と述べました。これを図 1 で考えてみたいと思います。

図 1 は、左右とも黒線が

$$y = f(x) = 3x^2 + 5x + 2$$

を表しています。図 1 左は、 $x = 1$ のとき、 x をほんの少しの値 (Δx) として 1 ($\Delta x = 1$) としたものです。 $x = 1$ のとき $f(1) = 10$ で、ほんの少し増やした値 $x + \Delta x = 2$ で、 $f(2) = 24$ です。このとき、 y の増分は $\Delta y = f(2) - f(1) = 14$ となります。図では赤で囲まれた 3 角形になります。このときの変化率は $\Delta y / \Delta x = 14 / 1 = 14$ となります。赤線と黒線は、多少乖離していて、 Δx の値を小さくすれば、赤線の傾きは、黒線に近づきそうです。

図 1 右は、 $x = 1$ のとき、 $\Delta x = 0.5$ としてみて、緑色の線が、 $\Delta x, \Delta y$ を表しています。黒線と緑線の乖離はだいぶ小さくなっていることがわかり、 $\Delta y / \Delta x = 6.25$ となっています。 Δx の値をどんどん小さくしていけば、 $x = 1$ のとき、 x を増やしたときの y の増分の割合が求まります。これを、1.4 小節で求めて見たいと思います。

1.3 微分のサイト (導関数を求めるサイト) の利用

微分してほしい数式を入力すると、自動で微分して、その結果の導関数を表示してくれます。ここでは、[Derivative Calculator • With Steps!](https://www.derivative-calculator.net/)(<https://www.derivative-calculator.net/>) を利用します。




図1 微分の考え方（理論値:黒線と点線で囲まれた面積 ⇒ 近似値:赤（緑）線と点線で囲まれた三角形の面積）

- (1) [Derivative Calculator • With Steps!](#)にいきます。
- (2) 「Calculate the Derivative of …」の下の入力欄に、微分したい数式を入力します
ここでは、 $y = f(x) = 3x^2 + 5x + 2$ を微分するので、

$$3x^2 + 5x + 2$$

と入力します。 x^2 乗は x^2 のように入力します。

(3) **Result** の下の **FIRST DERIVATIVE:** (一階微分) に、微分した結果の導関数 $f'(x) = 6x + 5$ が表示されます。

動画:導関数を求める

1.4 表計算で微分の近似値を求める

$y = f(x) = 3x^2 + 5x + 2$ ($f'(x) = 6x + 5$) の $x = 1$ のとき、 $\Delta y / \Delta x$ の Δx を 0 に近づけたときの値を図 1 の方法で求めたいと思います。求めたい値は、 $f'(x) = 6x + 5$ なので、 $f'(1) = 11$ であることは分かっているのですが、求める方法の考え方を学修するため表計算で、 $f'(1)$ の近似値を求めます。

図 2 のひな形は、テキストのページから、differential.xlsx というファイル名でダウンロードできます。

1.4.1 $x = 1$ で、 $\Delta x = 1$ の場合

図 2 の A5:G5 は、 $x = 1$, $\Delta x = 1$ のとき、 $\Delta y / \Delta x$ の値を計算しているものです。

- (1) A5 に $x = 1$ の 1 を入力します。
- (2) B5 に $f(x)$ の値を計算します。 $f(x) = 3x^2 + 5x + 2$ の式ですが、A5 に x の値があるのでそれを利用し、
 $=3*A5^2+5*A5+2$ と入力します。
- (3) C5 に Δx の値、1 を入力します。
- (4) D5 に $x + \Delta x$ の式、 $=A5+C5$ とします。
- (5) E5 に、 $f(x + \Delta x)$ の値を計算します。
 $=3*D5^2+5*D5+2$ と入力します。
- (6) F5 に $\Delta y = f(x + \Delta x) - f(x)$ の計算式、 $=E5-B5$ を入力します。
- (7) G5 に $\Delta y / \Delta x$ の計算式、 $=F5/C5$ を入力します。

	A	B	C	D	E	F	G	H	I	J	K
1			式	$y=3x^2+5x+2$					導関数		
2											
3		計算式 (D2)を 入力	ちょっと増 やす値 (xの増分)	ちょっと増や した値 (x+xの増分)	計算式 (D2)を入 力	xをちょっと増 やしたとき、y の増分	x 増分 (Δx) の何倍 y は増え るか			左の表か ら 求めた値	導関数(J1 の式)から 求めた値
4	x	$y=f(x)$	Δx	$x+\Delta x$	$f(x+\Delta x)$	$\Delta y=$ $f(x+\Delta x)-f(x)$	$\Delta y/\Delta x$		x	y'	y'
5	1	10	1	2	24	14	14		1	11.0059	11
6	1	10	0.5	1.5	16.25	6.25	12.5		2	17.0059	17
7	1	10	0.25	1.25	12.9375	2.9375	11.75		3	23.0059	23
8	1	10	0.125	1.125	11.42188	1.421875	11.375		4	29.0059	29
9	1	10	0.0625	1.0625	10.69922	0.69921875	11.1875		5	35.0059	35
10	1	10	0.03125	1.03125	10.34668	0.346679688	11.09375				
11	1	10	0.015625	1.015625	10.17261	0.172607422	11.046875				
12	1	10	0.0078125	1.0078125	10.08612	0.086120605	11.0234375				
13	1	10	0.0039063	1.00390625	10.04301	0.043014526	11.01171875				
14	1	10	0.0019531	1.001953125	10.0215	0.021495819	11.00585938				

図 2 表計算で微分の考え方の学修

1.4.2 $x = 1$ で、 Δx の値を変化させたとき、 $\Delta y/\Delta x$ の計算

A6:G14 に、 Δx の値を 0 に近づけていき、 $\Delta y/\Delta x$ の値の変化を計算します。

(1) A6 は、A5 に同じなので $\boxed{=A5}$ とします。

(2) B6 は、B5 と x の値が同じなので $f(x)$ の値も同じなので $\boxed{=B5}$ とします。

(3) C6 は、 Δx の値で 0 に近づけていきます。ここでは、上の行の値の 1/2 にしていきます。 $\boxed{=C5*0.5}$

(4) D6:G6 の値は、A6:C6 から計算されるので、上の行と計算式は同じです。

複写元: D5:G5 複写先: D6:G6

(5) 7 行目以降は、6 行目の計算式と同じです。

複写元: A6:G6 複写先: A7:G14




図3 Δx を変化させたときの $\Delta y / \Delta x$ の変化 ($x = 1$ の場合)

動画:[x = 1 のときの変化率を求める](#)

図3は、 Δx を変化させたときの $\Delta y / \Delta x$ の変化の散布図で、C4:C14 と F4:F14（複数の領域）を範囲指定して作成したものです。 Δx が0に近づくにつれ、 $\Delta y / \Delta x$ が11に収束する様子が確認できると思います（ Δx の値が小さくなる、左方

向に移動するにつれ、 $\Delta y/\Delta x$ が 11 に近づいていく様子).

[動画:収束散布図](#)

1.4.3 各 x とその $\Delta y/\Delta x$ 近似値の一覧

I5:J9 に、 x とその近似値の一覧を作成します.

- (1) x を 1, 2, ..., 5 と変化させ、その近似値を求めます. I5:I10 に、図 2 のように、1, 2, 3, 4, 5 と入力します.
- (2) $x = 1$ のときの収束値を求めます. A5 に 1 と入力します (すでにそうになっていると思いますが). G14 にその近似値が表示されますので、コピーして、値貼り付け で J5 に転記します.
- (3) $x = 2$ のときの収束値を求めます. A5 に 2 と入力します. G14 にその近似値が表示されますので、コピーして、値貼り付け で J6 に転記します.
- (4) 同様に、2, 3, 4, 5 の近似値を求めて、転記します.

次に、1.3 節の Web で求めた導関数から求めた理論値との比較を行います. $y = f(x) = 3x^2 + 5x + 2$ の導関数は、 $f'(x) = 6x + 5$ なので、その計算をします.

- (1) K5 に =I5*6+5 と入力します.
- (2) K5 の式を下に複写します.

複写元: K5 複写先: K6:K9

J 列の値は、K 列の値を近似していることが分かり、確認できました.

[動画:近似値一覧表の作成](#)

1.5 練習問題 (微分)

他の関数で表を作成し、理論値を近似していることを確認しましょう。例えば、 $y = f(x) = ax^3 + bx + c$ とし、 a を学績番号の 1 の位、 b を学績番号の 100 の位と 10 の位の 2 桁の値、 c は、西暦の年度の 1 の位とします。

2024 年度の授業で、学生番号の下 3 桁が 123 の場合、 $a = 3, b = 12, c = 4$ とし、 $y = f(x) = 3x^3 + 12x + 4$ について分析を行います。

分析は、すでに分析を行った differential.xlsx のコピーを作成して分析を行いましょう。

- (1) 1.3 節の Web で、導関数を求めます。
- (2) A5:G14 の計算式のうち、必要な部分を修正します。
- (3) I5 から I10 にチェックする x の値を決めます。
- (4) A5 にチェックする x を入力し、その結果を J5:J9 に転記していきます。
- (5) K5:K9 に (2) で求めた導関数の式で、 $f'(x)$ の値を求めます。
- (6) 図 3 のような散布図を作成して、理論値に収束していくかを確認します。

2 積分の概念

2.1 積分の意味

社会科学などでは，積分は，累積値を求めるときに使われます．例えば， $f(x)$ を機械起動から x 分後の製品の 1 分あたりの生産量の関数^{*1}とします．0 分目（機械起動時）から，10 分目までの生産量の累積値は，

$$\int_0^{10} f(x)dx$$

という式で，20 分目から 25 分目までは，

$$\int_{20}^{25} f(x)dx$$

という式で計算できます． $\int_a^b f(x)dx$ で， $x = a$ から $x = b$ までの $f(x)$ の値の累積値を求めることを意味します．

この式は，離散値の合計を計算する $\sum$ (Excel では sum 関数) に対応しており，0 分目から 1 分目までの 1 分間の生産量を x_1 ，1 分目から 2 分目までの 1 分間の生産量を $x_2, \dots$ 19 分目から 20 分目までの 1 分間の生産量を x_{20} とすると，それらの累積値は，

$$\sum_{i=1}^{20} x_i$$

という式で計算できます．積分も合計も共に累積値を計算できます．1 分単位で毎分の実産量を表現されている場合 (x_i を i 分目の生産量で， i は整数などの離散値) など離散的に表現されている場合，合計 ($\sum$) で計算します．ある瞬間の実産量を $f(x)$ (x は連続量， $x = 1.3456$ など可能) などの関数で表現できる場合，累積値は積分で求めることができます．

^{*1} $f(x)$ は， x 分後の瞬間の実産量を表します．生産の速度は変化しているので，1 分間は同じ速度で生産していると仮定して，その 1 分間の生産量を表しています

2.2 近似値を求める

$y = f(x) = x^2 + 2x + 3$ で、0 から 4 までの積分値 ($\int_0^4 f(x)dx$) の近似値を求めてみます。




図4 台形の面積で積分値の近似値を求める

図4 左の青の曲線は、 $y = f(x) = x^2 + 2x + 3$ を表します。 $x = 0$ から 4 までの積分値 ($\int_0^4 f(x)dx$) は、青の曲線と青の直線で囲まれた部分の面積になります。この面積を図4 右のようにオレンジの線で囲まれた $x = 0 \sim 2$ と $x = 2 \sim 4$ の2つの台形の面積の合計で近似して計算してみましょう。

$y = f(x) = x^2 + 2x + 3$ で, $f(0) = 3, f(2) = 11, f(4) = 27$ となります. 台形の面積は,

$$\text{台形の面積} = \frac{\text{上底} + \text{下底}}{2} \text{高さ}$$

です. 図では, 高さが X 軸での台形の幅, 上底と下底が $f(x)$ の値になっています. 台形 1 の面積 $= (3 + 11) * 0.5 * 2 = 14$, 台形 2 の面積 $= (11 + 27) * 0.5 * 2 = 38$ となり, その合計 52 が, 積分値の近似値になります. (次小節で求めるように, 理論値は, $49\frac{1}{3}$ になります.)

2.3 積分値を求める

$\int_0^4 f(x)dx$ のように, ある範囲 (0 から 4 まで) の面積を求めることは定積分と呼ばれています. 定積分を行うには, 通常, 不定積分関数 $F(x)$ を求めます. $F(x)$ は, 微分すると $f(x)$ になる関数です. 一般によく利用する関数であれば, 次小節のように Web で求めることができます. $f(x) = x^2 + 2x + 3$ の場合, $F(x) = \frac{x^3}{3} + x^2 + 3x + C$ で, C は積分定数と呼ばれ, 任意の C のとき, $F(x)$ を微分するとその導関数が $f(x)$ になるので, C を加えます. $C = 0$ とおいて処理することがあります.

定積分は次のように計算されます.

$$\int_a^b f(x)dx = | F(b) - F(a) |$$

$f(x) = x^2 + 2x + 3$ で, 0 から 4 までの不定積分結果は, $F(x) = \frac{x^3}{3} + x^2 + 3x + C$ より,

$$\int_0^4 f(x)dx = | F(4) - F(0) | = \frac{64}{3} + 16 + 12 + C - C = 49\frac{1}{3}$$

となります.

2.4 積分のサイト (関数を求めるサイト) の利用

積分してほしい数式を入力すると、自動で積分して、その結果の不定積分関数を表示してくれます。ここでは、[Integral Calculator • With Steps!](https://www.integral-calculator.com/)(<https://www.integral-calculator.com/>) を利用します。

- (1) [Integral Calculator • With Steps!](https://www.integral-calculator.com/)にいきます。
- (2) 「Calculate the Integral of ...」 の下の入力欄に、積分したい数式を入力します
ここでは、 $y = f(x) = x^2 + 2x + 3$ を積分するので、

$$x^2 + 2x + 3$$

と入力します。 x^2 乗は x^2 のように入力します。

- (3) Result の下の "MANUALLY" COMPUTED ANTIDERIVATIVE: に、不定積分関数 $F(x) = \frac{x^3}{3} + x^2 + 3x + C$ が表示されます。
- (4) 下方のグラフの右の「Table of values:」の $x =$ に入力すれば、 $F(x)$ の値が表示されます。ただし、 C は、Assign variables: で設定された値です。

[動画:積分のサイトの利用](#)

2.5 表計算で積分の近似値を求める

図 5 は、台形の面積を使って、積分値の近似値を求める表です。図 5 のひな形は、テキストのページから、integral.xlsx というファイル名でダウンロードできます。

$y = f(x) = x^2 + 2x + 3$ の 0 からの積分値 $\int_0^{10} f(x)dx$ の近似値を次の手順で計算します。 [動画:表計算を使って積分値の](#)

	A	B	C	D	E
1	式	$y=x^2+2x+3$			
2	Δx (刻み幅)	2	台形の面積の総和 470		
3		B1の計算式を入力	台形の幅	x を Δx 分増やしたとき(次の行)の $f(x)$	台形の面積
4	x	b: $y=f(x)$	a: Δx	c: $f(x+\Delta x)$	$0.5a(b+c)$
5	0	3	2	11	14
6	2	11	2	27	38
7	4	27	2	51	78
8	6	51	2	83	134
9	8	83	2	123	206
10	10	123	2	0	




図5 表計算で積分の近似値 ($\Delta x = 2$)

理論値:青線とうすだいだい色で囲まれた面積 $\Rightarrow$ 近似値:オレンジとうすだいだい色で囲まれた台形の面積の総和

計算 (図5の説明)

- (1) Δx (台形の高さ, だんだん小さくしていく)を決め, B2に入力します. 最初は2にします.
- (2) A5に, 最初の x の値0を入力します.
- (3) B5に, $f(x)$ の計算式を入力します.

$$\boxed{\text{B5:}} = A5^2 + 2 * A5 + 3$$

(4) C5 に Δx の値を B2 から転記します.

$$\boxed{\text{C5:}} = B2$$

(5) D5 に, $f(x + \Delta x)$ の値を入れますが, これは, 次の行の $f(x)$ の値です (次の行の式を設定したら正しい値が表示されます).

$$\boxed{\text{D5:}} = B6$$

(6) E5 に, 台形の面積を計算します. 高さは Δx 上底は $f(x)$, 下底は $f(x + \Delta x)$ です.

$$\boxed{\text{E5:}} = 0.5 * C5 * (B5 + D5)$$

(7) A6 は, 次の x の値, 前の $x + \Delta x$ です.

$$\boxed{\text{A6:}} = A5 + C5$$

(8) B6 は, B5 と同じ計算式です.

複写元	B5	複写先	B6
-----	----	-----	----

(9) C6 の計算式は, C5 の値と同じ値です.

$$\boxed{\text{C6:}} = C5$$

(10) D6:E7 は, D5:E6 と同じ計算式です.

複写元 D5:E5 複写先 D6:E6

(11) 6 行目を下方に複写します.

複写元 A6:E6 複写先 A10:E10

(12) $x = 10$ の行の台形の面積は, $x = 10 \sim 12$ の台形の面積で不要なので, E10 の値を削除します.

(13) E2 に台形の面積の総和を計算し, それを積分値の近似値とします.

E2: =SUM(E5:E9)

動画:積分の近似値の計算 ($\Delta x = 2$)

図 6 は, $\Delta x = 2$ の表を複写, 修正して, $\Delta x = 1$ の計算表を作成したものです.

動画:積分の近似値の計算 ($\Delta x = 1$)

Web を使って, $f(x) = x^2 + 2x + 3$ の $F(10) - F(0)$ は, 463.333 となり, $\Delta x = 2$ の 470, $\Delta x = 1$ の 465 と 463.333 に近づいていくのがわかります.

2.6 練習問題 (積分)

他の関数, たとえば, 1.5 節の練習問題 (微分) と同じ式を用います. 計算表は, integral.xlsx をコピーして利用します.

(1) 2.4 節の Web で, $F(10) - F(0)$ の値を求めます.

(2) $\Delta x = 2$ の表の A5:E10 の計算式のうち, 必要な部分を修正します.

(3) $\Delta x = 1$ の表の A5:E15 の計算式のうち, 必要な部分を修正します.

(4) 2 つの表の E2 の値が (2) の $F(10) - F(0)$ の値に近似しているのか確かめます.

	A	B	C	D	E
1	式	$y=x^2+2x+3$			
2	Δx (刻み幅)	1	台形の面積の総和		465
3		D2の計算式を入力	台形の幅	xを Δx 分増やしたとき(次の行)の $f(x)$	台形の面積
4	x	b: $y=f(x)$	a: Δx	c: $f(x+\Delta x)$	$0.5a(b+c)$
5	0	3	1	6	4.5
6	1	6	1	11	8.5
7	2	11	1	18	14.5
8	3	18	1	27	22.5
9	4	27	1	38	32.5
10	5	38	1	51	44.5
11	6	51	1	66	58.5
12	7	66	1	83	74.5
13	8	83	1	102	92.5
14	9	102	1	123	112.5
15	10	123	1		
16					
< > 積分 $\Delta x=2$ <u>積分$\Delta x=1$</u> 積分 $\Delta x=0.1$ 積分 $\Delta x=0.01$ Sheet1 Sheet2 SI					

図6 表計算で積分の近似値を求める ($\Delta x = 1$)

3 指数・対数

3.1 一次関数・指数関数とは

人文社会科学ではどんな関数が分析に使われるのでしょうか？ 著者の見ている範囲で基本的な分析で使われる関数は、

- 一次関数, $y = f(x) = 3x + 10$ のような関数
- 指数関数, $y = f(x) = 1.2^x$ のような関数 (とその逆関数である対数関数)

の 2 つです.

一次関数は, たとえば 50 円値引きのクーポン券を持っているとき, 1 個 120 円のリングを x 個購入したときの支払金額は,

$$y = 120x - 50$$

で表現できます. 一次関数は, x が 1 増えると y は同じ値増えるという関係 (一定数で増加) を表現しています.

一方, 指数関数は, 10 万円借りたとして, 毎年 1.2 倍になる借金 (20% の利息) を全く返済しなかったとして, x 年後の借金の金額 (元利合計額) y は

$$y = 10 \times 1.2^x$$

で表現できます. 指数関数は, x が 1 増えると y は同じ倍数増えるという関係 (一定倍で増加) を表現しています.

- 一次関数, $y = ax + b$, x が 1 増えると y は a 増える関係
- 指数関数, $y = f(x) = b \times a^x$, x が 1 増えると y は a 倍になる関係

を表します. 図 7 は, 一次関数 $y = 10x + 10$, 指数関数 $y = 10 \cdot 2^x$ のグラフです. 1 次関数は x が 1 増えると y は 10 増え




図7 一次関数 $y = 10x + 10$, 指数関数 $y = 10 \cdot 2^x$ のグラフ

る関係を指数関数は x が 1 増えると y は 2 倍になる関係を示しています。

本節では、指数・対数関数の基本的な使い方と表計算ソフトウェアでの計算方法を次節では、複数の一次関数をまとめて表計算ソフトウェアでの計算する方法を学修します。

3.2 表計算で指数・対数の利用

2つの例題で考えていきたいと思います。

例題 1:

ある感染症の国内感染者数は、毎日 1.2 倍 ($1.2 - 1 = 0.2 = 20\%$ 増加) になります。ある日 (0 日目) の感染者を 10 人とします。 $x(> 0)$ 日後に感染者数を y とします。

例題 2:

ある貸金業者の利率 (年利) は、15% です。ある日 (0 年目) に、100 万円借りたとしたし、(返済をしないとして、) $x(> 0)$ 年後の借り入れ額 (元利合計額) を y とします。

両例題とも指数関数

$$y = b \cdot a^x$$

という関数であらわすことができます。 b は、 $x = 0$ のときに y の値で、例題 1 の場合、10 になります。 a は、 x が 1 増加したとき、 y が何倍になるのかを表す値で、例題 1 では、1.2 倍になるので、 $a = 1.2$ になります。 そうすると、例題 1 は次式になります。

$$y = 10 \cdot 1.2^x \quad (\text{例題 1})$$

例題 2 では、0 年目は 100 (元金) ですので、 $b = 100$ 、毎年 15% の利息は、毎年、1.15 倍になるので、 $a = 1.15$ となり、次式になります。

$$y = 100 \cdot 1.15^x \quad (\text{例題 2})$$

以下の小節では、Excel のファイル、exp_log.xlsx を使います。テキストのページからダウンロードできます。

3.2.1 指数関数で表現 (a, b, x から y を求める)

図 8 の A2:E4 のように、 a, b, x から y を求めます。 B4:D4 に、例題 1 の数値を入力します。 E4 は、 $b \cdot a^x$ ですので、次のように入力します。

	A	B	C	D	E	F	G	H	I
1	例題 1								
2	yを求める								
3		a	b	x	y:計算式				
4		1.2	10						
5	xを求める								
6		a	b	y	何倍	x:計算式(対数)		(1a)試行錯誤	
7								(1b)ゴールシーク	
8								(1c)計算式	
9	aを求める								
10		y	b	x	y/b	a:計算式(x乗根)		(2a)試行錯誤	
11								(2b)ゴールシーク	
12								(2c) 計算式	

図 8 表計算で指数・対数の計算

$$\boxed{\text{E4:}} = \text{C4} * \text{B4}^{\text{D4}}$$

$\boxed{\wedge}$ は、べき乗を表します。 x に 10 を設定すれば、 y は、 61.9 くらいになります。

3.2.2 対数, 試行錯誤, ゴールシーク (a, b, y から x を求める)

例題 1 で感染者数が 1000 人になるのが、何日後かを求めたいと思います。

■**試行錯誤** $\text{D4}(x)$ の値を適当に (試行錯誤で) 変化させ、 $\text{E4}(y)$ が 1000 になる値を求めます。 x が 25 くらいのとき、だいたい 1000 になると思います。この作業をするとき、皆さんは、 x を大きくすれば y は大きくなるという性質を直感的に利用

していたと思います。これは、日数 (x) が増えれば、毎日 1.2 倍になるので、感染者数も増えるという常識にも一致します。このように x が増えれば、 y が増えるという関係の関数を単調増加関数といいます。この性質を利用して、求めたものです。

動画:試行錯誤で x を求める

■ゴールシーク 表計算ソフトウェアにはこのような試行錯誤を自動的に実行してくれる機能があり、ゴールシーク^{*2}と呼ばれています。

(1) メニューの データ → 予測 → What-If 分析 → ゴールシーク

(2) ゴールシークの設定

- 数式入力セル:結果 (y) の計算式が設定されているセル, この場合 E4
- 目標値:結果 (y) の計算式にとってほしい値, この場合 1000 にしたいので, 1000 を入力
- 変化させるセル: x を変化させて, y を求めるので, x のセル D4 を指定

(3) OK をクリックし, しばらく計算すると結果が D4:E4 に表示されます。

動画:ゴールシークで x を求める

■計算式 (対数) で求める x から y を求めるには指数関数 ($y = a^x$) を利用しました。 y から x を求めるには指数関数の逆関数である対数関数 ($x = \log_a y$) を利用します。

$$y = a^x \quad \leftrightarrow \quad x = \log_a y$$

^{*2} ゴールシークではうまく求めることができない場合があります。うまく求めることができないかは、対象の関数の性質によります。これは、利用する関数の学修になります。指数関数 $y = a^x$ で、 $a > 0, x \geq 0$ の場合、狭義単調増加関数という、 x が増加すればかならず y も増加するという性質があるので、 y から x の値をゴールシークで求めることができます。

例題の場合, $y = b \cdot a^x$ なので.

$$y = b \cdot a^x \quad \leftrightarrow \quad x = \log_a \frac{y}{b}$$

となります. B6:F7 の計算式は, 次のように設定します.

$$\boxed{\text{E7:}} = \text{D7} / \text{C7} \quad \boxed{\text{F7:}} = \text{LOG}(\text{E7}, \text{B7})$$

Excel の関数で, $\log_a y$ は, $=\text{LOG}(\text{y のセル}, \text{a のセル})$ とします.

3 つ方法で計算した値は近い値になっていることが分かります.

3.2.3 n 乗根で求める (b, x, y から a を求める)

例題 1 で感染者数が 500 人になったのがのが 20 日後だったとします. a (毎日何倍になったか) はいくつかを求めます.

「 a, b, y から x を求める」と同様に, 試行錯誤やゴールシークで求めることができます. ゴールシークでは, 次のように設定します.

- 数式入力セル:E4
- 目標値:500
- 変化させるセル: B4 (a のセル)

動画:[ゴールシークで \$a\$ を求める](#)

計算式では, 次のようになります.

$$\begin{aligned} y &= b \cdot a^x \\ \frac{y}{b} &= a^x \\ a &= \sqrt[x]{y/b} = \left(\frac{y}{b}\right)^{\frac{1}{x}} \end{aligned}$$

式の2行目のように、 a を x 乗したら y/b になる a を探すことです。これは、式の2行目のように、 y/b を $1/x$ 乗した値になります。B10:F11の式は次のようになります。

$$\boxed{\text{E11:}} = \text{B11}/\text{C11} \quad \boxed{\text{F11:}} = \text{E11}^{(1/\text{D11})}$$

3.2.4 a, x, y から b を求める

例題1で感染者数が500人になったのが10日後だったとします。 b (0日目の感染者数)はいくつか求めます。これは、次の計算式で簡単に求まります。

$$y = b \cdot a^x$$

$$b = \frac{y}{a^x}$$

B14:F15の式は次のようになります。

$$\boxed{\text{E15:}} = \text{D15}^{\text{B15}} \quad \boxed{\text{F15:}} = \text{C15}/\text{E15}$$

3.3 金利計算

例題2と同様に、金利計算でも指数関数が使われます。金利計算では、

$$F = P \cdot (1 + r)^n$$

という式で表現されます。 P は現在価値(元金)、 F は将来価値(元利合計)、 r は利率(年単位であれば年利)、 n は期間(年単位であれば年数)を意味します。指数関数 $y = ba^x$ との対応では、 $F = y, P = b, (1 + r) = a, x = n$ となります。

$F = P \cdot (1 + r)^n$ で、たとえば、年利 5% の利率と期間 5 年を与えれば、現在の金額（現在価値）万円を将来の金額（将来価値）に変換できます。 $F = 100 \cdot (1 + 0.05)^5 = 100 \cdot 1.276 = 127.6$ 万円となります。

指数関数と同様に、 F, P, r, n (すべて正の値) のうち、3 つの値が分かれば、他の 1 つの値を求めることができます。

指数関数を使って計算する方法は、複利計算と呼ばれています。しかし、実務では、 n の整数以外の計算は単利計算で行います。^{*3} 本授業では単純に指数関数を利用します。

3.3.1 練習問題

以下の問題を複利計算（指数関数）で求めよ。例題 1 のワークシートをコピーして例題 2 のワークシートを作成し、適当に変更して、解答のワークシートを作成せよ。金額は万円単位で計算せよ（100 万円は 100 とする）。注意： $a = 1 + r$

※ 2024/11/18 (3),(4),(5) の現在価値と年利を指定しました。授業等で指定された値があればそれを利用してください。

- (1) 現在価値 100, 年利 15%, 期間 7.3 年とする。将来価値 F を求めよ。
- (2) 現在価値 100, 年利 0.5%, 期間 7.3 年とする。将来価値 F を求めよ。
- (3) F が 300 になるのが何年後か (n) を試行錯誤で求めよ（現在価値 100, 年利 1.3%）。
- (4) F が 300（万円）になるのが何年後か (n) をゴールシークで求めよ（現在価値 100, 年利 1.3%）。
- (5) F が 300（万円）になるのが何年後か (n) を計算式を設定して求めよ（現在価値 100, 年利 1.3%）。
- (6) 年利 r が不明とします。現在価値が 100, 11.3 年後の将来価値が 1000 とします。試行錯誤で r をもとめよ。

※ ヒント： a を求め、 $a = (1 + r)$ で r を求めます。

- (7) 年利 r が不明とします。現在価値が 100, 11.3 年後の将来価値が 1000 とします。ゴールシークで r をもとめよ。
- (8) 年利 r が不明とします。現在価値が 100, 11.3 年後の将来価値が 1000 とします。Excel の計算式で r をもとめよ。
- (9) P (現在価値, 元金) が不明とし、年利 15% で 12.5 年後将来価値が 15 でした。 P を Excel の計算式で求めよ。

^{*3} 例えば 3.4 年の場合、3 年分は指数関数（複利計算）で、0.4 年分は単利計算（一次関数）で計算します。整数部分とそれ以外の部分を場合分けをして計算します。昔は、PC や関数電卓がなく、小数を含む指数・対数計算が簡単に行うことができないという経緯があるのだと思います。

4 表計算によるベクトル・行列の計算

4.1 ベクトル表現

ベクトルは、 n 次元の数の組を表すことができます。例えば、情報入門 2 の第 2 章のページ (2-2) の図 1，を各都道府県の米，パン，ケーキ，せんべい，チョコレートの値をベクトルで表すと次のようになります。

$$\boldsymbol{a}_1 = (27371, 27609, 8339, 4427, 6731)$$

←札幌市のベクトル

$$\boldsymbol{a}_2 = (20934, 25662, 6563, 5955, 4861)$$

←青森市のベクトル

$$\boldsymbol{a}_3 = (22927, 28638, 7984, 6249, 5678)$$

←盛岡市のベクトル

(米，パン，ケーキ，せんべい，チョコレート) の値

情報入門 2 第 2 章の 10.1.1 のクラスタリングで，サンプル間の距離を求めることは，対象の 2 つの都市のベクトルの距離（ユークリッド距離）を求めることになります。

このように，各サンプルをベクトルで表現することはよく行われます。

4.2 ベクトルの積（内積）

4.2.1 ベクトルと内積

ベクトルや行列は、どのように使うのでしょうか？ おもな利用法は、1 次式（行列は複数の 1 次式）を表現するのに使います。

簡単な例：リンゴ 2 個，ミカン 5 個，バナナ 12 本を購入します。リンゴ 1 個 x_1 円，ミカン 1 個 x_2 円，バナナ 1 本 x_3 円としたとき，いくらになるのでしょうか？ その金額を y とすると

$$y = 2x_1 + 5x_2 + 12x_3$$

となります。これをベクトルで表すと，

$$\boldsymbol{a} = (2, 5, 12)$$

←数量の行ベクトル

$$\boldsymbol{x} = \begin{pmatrix} x_1 \\ x_2 \\ x_3 \end{pmatrix}$$

←価格の列ベクトル

となります。(2, 5, 12) の 2 は第 1 成分，5 は第 2 成分と呼ばれます。また，成分の数 3 は，一般に次元と呼ばれます。

金額の計算は，ベクトル $\boldsymbol{a}$ と $\boldsymbol{x}$ の内積 ($\cdot$) で表現できます。

$$y = \boldsymbol{a} \cdot \boldsymbol{x} = (2, 5, 12) \cdot \begin{pmatrix} x_1 \\ x_2 \\ x_3 \end{pmatrix} = 2x_1 + 5x_2 + 12x_3$$

内積は，2 つのベクトルの 1 番目の成分同士の積，2 番目の成分同士の積，... の合計 (積和) となります。内積を計算するにあたって，2 つのベクトルの次元数は同じでないと計算できません。

4.2.2 ベクトルと内積，表計算での計算

本書のホームページにある `VecMatrix.xlsx` を利用します．ベクトルの内積の計算は，ワークシート「内積」にあり，図 9 のように行います．図 9 では，リンゴ 200 円 ($= x_1$)，ミカン 50 円 ($= x_2$)，バナナ 120 円 ($= x_3$) の場合を計算しています．

	A	B	C	D	E	F	G	H
1	内積の計算							
2	3次元の場合							
3		a=	2	10	15			
4								
5		x=	200					
6			50			y=	2700	
7			120					
8								
9	4次元の場合							
10		a=	2	10	15	18		
11								
12		x=	5					
13			8			y=		
14			6					
15			7					

図 9 内積の計算

G6 は，内積の計算の Excel の関数です．

G6: `=MMULT(C3:E3,C5:C7)`

MMULT は、次小節で説明する行列とベクトルの積などを計算する関数ですが、ベクトルの内積も計算できます。他に、sumproduct という関数がありますが、これは積和計算の関数で、Excel の 2 つの行（または 2 つの列）の各セル同士の積和になり、図のように、行と列の積和には対応していません。（sumproduct を使うのであれば、両方とも行にするか、両方とも列にする必要があります。）

リンゴ、ミカン、バナナの単価を変更したり、購入数量を変更してみて、計算が正しく行われているかどうか確かめましょう。

4.2.3 練習問題

内積のワークシートの G13 に、4 次元の行ベクトル列ベクトルの内積の計算結果を表示するようにしよう。

また、リンゴ (@170)、ミカン (@85)、バナナ (@100)、パイナップル (@530) として、リンゴ 5 個、ミカン 12 個、バナナ 7 本、パイナップル 1 個購入したときの金額をこの行列を使ったワークシートで計算しましょう。

※「@」は単価を表します。

4.3 行列、行列とベクトルの積、逆行列

4.3.1 行列とは

前小節の例は、1 人の人の購入個数で、計算しました。本小節では、1 番、2 番、3 番の人が、

人	リンゴ	ミカン	バナナ
1 番	1	3	4
2 番	3	5	4
3 番	6	3	7

購入します。それぞれの人の購入金額のベクトルを求めてみたいと思います。それぞれの単価は前小節の例と同じとします。

この場合、数量の表を行列という表現法で表します。この行列を $\mathbf{A}$ とします。

$$\mathbf{A} = \begin{pmatrix} 1 & 3 & 4 \\ 3 & 5 & 4 \\ 6 & 3 & 7 \end{pmatrix}$$

行列 $\mathbf{A}$ の 1 行目が 1 番の人の購入数量、2 行目が 2 番の人の購入数量、3 行目が 3 番の人の購入数量に対応します。このように、3 人の行ベクトルを縦に並べた形になっています。

一般に行列は、

$$\begin{pmatrix} a_{11} & a_{12} & \dots & a_{1n} \\ a_{21} & a_{22} & \dots & a_{2n} \\ \vdots & \vdots & \ddots & \vdots \\ a_{m1} & a_{m2} & \dots & a_{mn} \end{pmatrix}$$

のような形をしており、行数、列数が同じである必要はありません。

4.3.2 行列とベクトルの積

前小節の例題で、金額は、購入数量の行ベクトル (a) と価格の列ベクトル (x) の内積で計算しました。

$$y = \mathbf{a} \cdot \mathbf{x}$$

それぞれの成分毎の積を求め、その合計でしたので、行ベクトル (a) の次元と価格の列ベクトル (x) の次元は、等しくなる必要があります。

行列の場合、行列 $\mathbf{A}$ とベクトル x の積で、金額の列ベクトル $\mathbf{y}$ を求めます。

$$\mathbf{y} = \mathbf{Ax} = \begin{pmatrix} 1 & 3 & 4 \\ 3 & 5 & 4 \\ 6 & 3 & 7 \end{pmatrix} \begin{pmatrix} 200 \\ 50 \\ 120 \end{pmatrix} = \begin{pmatrix} 1 \times 200 + 3 \times 50 + 4 \times 120 \\ 3 \times 200 + 5 \times 50 + 4 \times 120 \\ 6 \times 200 + 3 \times 50 + 7 \times 120 \end{pmatrix} = \begin{pmatrix} 830 \\ 1330 \\ 2190 \end{pmatrix}$$

$\mathbf{y}$ の 1 行目は、1 番の人の行ベクトル (1, 3, 4) と価格の列ベクトル $\mathbf{x}$ との内積で求めています。2 行目は 2 番の人の行ベクトル、3 行目は 3 番の人の行ベクトルと価格の列ベクトルの内積になっています。

行列とベクトルの積の場合、内積が 2 つのベクトルの次元が等しいことが必要であるのと同様に、行列の列数（くだもの種類数）とベクトルの次元（くだもの種類数）は同じである必要があります。ただし、行列の行数にはそのような制約はありません。何人の人の金額の計算も表現できます。

4.3.3 行列とベクトル積，表計算での計算

図 10 の上方 2 行目から 8 行目に、Excel での行列の積の計算を行いました。

H6 から H8 に、ベクトル $\mathbf{y}$ を計算しています。

$$\boxed{\text{H6:}} \quad =\text{MMULT}(\text{C2:E4}, \text{C6:C8})$$

H7, H8 は、H6 の計算式で自動で拡張（スピル）されて表示されます。

4.3.4 逆行列

ベクトルの内積や行列のかけ算と似た問題で、実数（スカラー）の問題を考えてみたいと思います。 a をリンゴの個数（ただし、 $a \neq 0$ ）、 x をリンゴの単価、 y を金額とします。 x から y を求める式と y から x を求める式は次のようになります。

$$y = ax$$

$$x = \frac{1}{a}y = a^{-1}y$$

	A	B	C	D	E	F	G	H	I	J	K	L	M
1													
2			1	3	4								
3	A=		3	5	4								
4			6	3	7								
5													
6			200					830		←行列Aの1行目			
7	x=		50			y=		1330		←行列Aの2行目			
8			120					2190		←行列Aの3行目			
9													
10			-0.44231	0.173077	0.153846			830			200		
11	A^{-1}=		-0.05769	0.326923	-0.15385	y=		1330	A^{-1}y=		50	← x に等しい	
12			0.403846	-0.28846	0.076923			2190			120		
13													
14			1	2.22E-16	1.11E-16								
15	A^{-1}A=		-2.2E-16	1	0								
16			0	0	1								

図 10 行列とベクトルの積，逆行列の計算

ただし， $a^{-1} = \frac{1}{a}$ で， a^{-1} は， a の乗法の逆元と呼ばれています。

これらの式を使えば，3個(a)のリンゴを単価120(x)で購入した時の金額は360円($y = ax$ で計算)を求めることができ，5個(a)のリンゴの金額450円(y)のとき，単価は90円($x = a^{-1}y$ で計算)を求めることができます。行列も同様の計算法で，ベクトル x や y を求めることができます。

$$y = Ax$$

$$x = A^{-1}y$$

として求めることができます。 $\mathbf{A}^{-1}$ は、 $\mathbf{A}$ の逆行列と呼ばれ、前記の逆元に対応します。これは、 $\mathbf{A}$ の逆行列 $\mathbf{A}^{-1}$ を求めれば、 $\mathbf{y}$ から $\mathbf{x}$ を求めることができることを意味します。

これは、連立一次方程式

$$y_1 = 1x_1 + 3x_2 + 4x_3$$

$$y_2 = 3x_1 + 5x_2 + 4x_3$$

$$y_3 = 6x_1 + 3x_2 + 7x_3$$

で、 y_1, y_2, y_3 の値を与えれば、 x_1, x_2, x_3 を求めることができることに対応します。連立一次方程式の解を求める場合、 x の数と式の数と同じであることや、問題によっては、解無しということがあります。

逆行列を求めることとの対応では、 x の数と式の数と同じであることは、 $\mathbf{A}$ の行数と列数は同じであることに対応し、「解無しではない」という条件は、行列が一次独立という条件に対応します。一次独立という概念は、数学の授業で学修するでしょう。

4.3.5 逆行列, 表計算での計算

図 10 の 10 行目から、逆行列を求め、その逆行列から $\mathbf{x}$ を求めます。

図 10 の C10:E12 に、C2:E4 の行列の逆行列を求めます。逆行列の Excel での計算は、MINVERSE という関数を使います。

$$\boxed{\text{C10:}} \quad =\text{MINVERSE}(\text{C2:E4})$$

C10:E12 の他のセルはスピルで表示されます。

C10:E12 の逆行列 ($\mathbf{A}$) と $\mathbf{y}$ を使い、 $\mathbf{x} = \mathbf{A}\mathbf{y}$ で、 $\mathbf{x}$ を K10:K12 に求めます。

$$\boxed{\text{K10:}} \quad =\text{MMULT}(\text{C10:E12}, \text{H10:H12})$$

C6:C8 の x と等しくなったと思います.

4.3.6 練習問題

4 番の人が, リンゴ 5 個, みかん 3 個, バナナ 1 本購入したとします. ワークシート 行列 をコピーして利用し, そこに, 1 番から 4 番の人の行列などを作成し, その金額のベクトルを計算する表を作成しなさい.

※ 変数の数 (3) と式の数 (4) が異なるので, この練習問題の行列は逆行列を持ちません.

4.4 行列とベクトルの積を使ったシミュレーション (マルコフ連鎖)

ここでは, 行列とベクトルの積の応用例として, マルコフ連鎖を使ってシミュレーションを行います. 本節でのマルコフ連鎖の例題では, 多数の人がおり, その人はいくつかの状態があり, 毎日, 確率により状態が変化するモデルを扱います. 変化の確率を行列で表し, ある日の各状態の人数をベクトルで表し, 行列とベクトルの積で, 次の日の状態のベクトルを求めます.

4.4.1 例題 (架空の学食)

ある大学の学食では, 毎日全学生 (1000 人) が登校するという前提で, 昼食を無料で提供しています. 定食は, 3 種類, F 定食 (揚げ物定食), M 定食 (肉料理), V 定食 (野菜定食) があります. 初日は, なんらかの方法でどれかの定食の食券が配布されます. 食後に, その日に食べた定食毎にクジが用意され, そのクジにより次の日の食券が配布されます. 各定食毎の次の日の食券の確率は, 固定されています.

各定食毎の確率は, 表 1 のようにしました.

初日 (0 日目) は, F 定食に 400 人, M に 300 人, V に 300 が配布されるとします. どのように, 変化するのか 20 回 (20 日目まで) 計算してみましょう.

表 1 ある食堂のクジの確率

		ある日の定食 (元の状態)		
		F 定食	M 定食	V 定食
次の日の定食 (次の状態)	F 定食	10%	40%	10%
	M 定食	50%	30%	70%
	V 定食	40%	30%	20%
計		100%	100%	100%

表 1 の見方ですが, F 定食の食券を持っている人が翌日の食券のクジを引くと, 10% の確率で F 定食の食券が, 50% の確率で M 定食の食券が, 40% の確率で V 定食の食券が配布されることを意味します. どれかの食券が配布されるので, 合計は 100% になります. 図 11 は, 状態の推移を表したものです.

4.4.2 マルコフ連鎖の計算

図 12 の B10:W13 のように, 各日の各食券の数がどのように変換するのかを計算します. 第 k 日目の各食券の数を列ベクトル $\boldsymbol{x}^k$ とします. 第何日目かは上付き文字 (肩字) で表現します. 第 0 日目のベクトルは, F 定食に 400 人, M に 300 人, V に 300 ですので次のようになります.

$$\boldsymbol{x}^0 = \begin{pmatrix} x_1^0 \\ x_2^0 \\ x_3^0 \end{pmatrix} = \begin{pmatrix} 400 \\ 300 \\ 300 \end{pmatrix}$$




図 11 例題の状態の推移

第 1 日目の F 定食は，F 定食の人の 10%，M 定食の人の 40%，V 定食の人の 10% に，F 定食の食券が配布されるので，次の式になります．同様に，第 1 日目の M,V 定食も計算できます．

$$x_1^1 = 0.1x_1^0 + 0.4x_2^0 + 0.1x_3^0$$

$$x_2^1 = 0.5x_1^0 + 0.3x_2^0 + 0.7x_3^0$$

$$x_3^1 = 0.4x_1^0 + 0.3x_2^0 + 0.2x_3^0$$

これを行列とベクトルの積の計算式で表現すると，次式になります．

$$\mathbf{x}^1 = \begin{pmatrix} x_1^1 \\ x_2^1 \\ x_3^1 \end{pmatrix} = \begin{pmatrix} 0.1 & 0.4 & 0.1 \\ 0.5 & 0.3 & 0.7 \\ 0.4 & 0.3 & 0.2 \end{pmatrix} \begin{pmatrix} x_1^0 \\ x_2^0 \\ x_3^0 \end{pmatrix} = \mathbf{A}\mathbf{x}^0$$

ただし,

$$\mathbf{A} = \begin{pmatrix} 0.1 & 0.4 & 0.1 \\ 0.5 & 0.3 & 0.7 \\ 0.4 & 0.3 & 0.2 \end{pmatrix}$$

とします. これは, 表 1 を行列の形にしたものです.

第 2 日目 $\mathbf{x}^2$ は, 行列 $\mathbf{A}$ と第 1 日目 $\mathbf{x}^1$ の積から求められます. 第 3 日目 $\mathbf{x}^3$ は, 行列 $\mathbf{A}$ と第 2 日目 $\mathbf{x}^2$ の積から求められます. したがって, 第 k 日目 $\mathbf{x}^k$ は, 行列 $\mathbf{A}$ と第 2 日目 $\mathbf{x}^{k-1}$ の積から求められます.

$$\mathbf{x}^1 = \mathbf{A}\mathbf{x}^0, \quad \mathbf{x}^2 = \mathbf{A}\mathbf{x}^1, \quad \mathbf{x}^k = \mathbf{A}\mathbf{x}^{k-1},$$

図 12 の B10:W13 では, このような繰り返しの計算が行われています.

4.4.3 Excel での計算

図 12 の表は, VecMatrix.xlsx のワークシート「マルコフ連鎖」にあります. 図 12 の C4:E6 は, 行列 $\mathbf{A}$ になります. また, C11:C13 は, 初日の食券数を与えています. これらの青のセルは, シミュレーションで変更してみる領域で, オレンジのセルは, 計算式を設定する領域です. 今回は, 20 日目まで計算してみます.

D11:D13(ベクトル $\mathbf{x}^1$) に, C11:C13(ベクトル $\mathbf{x}^0$) と C4:E6(行列 $\mathbf{A}$) の計算結果を表示し, $\mathbf{x}^2, \mathbf{x}^3, \dots, \mathbf{x}^{20}$ も同様な計算式などで複写します.

D11 に, 行列とベクトルの積の計算式を設定し, D11:D13 にその結果を表示します. この計算式を右に複写しますが, 行列の部分は, いつも同じセル範囲なので, 絶対参照を使います.

(1) 絶対参照を使って, D11 の計算式を設定します.

D11:	=MMULT(\$C\$4:\$E\$6,C11:C13)
------	-------------------------------

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S	T	U	V	W
1	推移確率行列																						
2			元の状態																				
3			F	M	V																		
4		F	10%	40%	10%																		
5	次の状態	M	50%	30%	70%																		
6		V	40%	30%	20%																		
7																							
8		検証	100%	100%	100%																		
9																							
10		何日目	0	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18	19	20
11		F	400	190	250	238.6	239.6	239.9	239.7	239.7	239.7	239.7	239.7	239.7	239.7	239.7	239.7	239.7	239.7	239.7	239.7	239.7	239.7
12		M	300	500	462	465.2	466.2	465.6	465.8	465.7	465.8	465.8	465.8	465.8	465.8	465.8	465.8	465.8	465.8	465.8	465.8	465.8	465.8
13		V	300	310	288	296.2	294.2	294.5	294.5	294.5	294.5	294.5	294.5	294.5	294.5	294.5	294.5	294.5	294.5	294.5	294.5	294.5	294.5

図 12 マルコフ連鎖の計算

- (a) D11 を選択し、「=MMULT(」 と入力し、
 - (b) C4:E6 を 範囲指定をし、F4 キーを押す、\$C\$4:\$E\$6 のように変化する（絶対参照の指定）。
 - (c) 「,」 を入力し、C11:C13 を範囲指定し、
 - (d) 「)」 を入力し、Enter
- (2) D11:D13 に、計算結果が表示されます。
- (3) D11:D13 の計算式を右に複写します。

複写元: D11:D13

複写先: W11:W13

動画:マルコフ連鎖の計算




図 13 各状態（各食券の枚数）の変化

図 13 のように，各期の食券の枚数の変化のグラフを作成しましょう．

4.4.4 シミュレーション

青のセルの部分の数値を変えて，どのように変化するか試してみましょう（シミュレーション）．

■回数の変化と食券数を変化

回数を増やしていくと、この $(240, 466, 295)^t$ ^{*4} くらいに近づいて行っていると思います。このように、ある値やベクトルに近づくことを収束すると言います。マルコフ連鎖では、多くの場合、あるベクトルに収束し、それを定常状態と言います。

■初期の食券数を変えてみる

C11:C13 の数値を変えて、第 20 日目 (W11:W13) がどのように変化するかシミュレーションをしていきましょう。ただし、C11:C13 の合計は 1000 で、各セルの値は非負の (0 以上) の整数とします。

合計が 1000 のとき、非負の整数であれば、おおよそ、 $(240, 466, 295)^t$ に近づいていくと思います。

■行列 A を変えてみる

C4:E6 の行列を変更してみましょう。ただし、各セルの値は、確率なので、0 以上 1 以下で、また、各食券のクジで、どれかの食券が 1 枚配布されるので、列方向の合計は、1(100%) です (C8:E8 の表示で確認できます)。多くの場合、 x^k の値は収束していくことが確認できると思います。

4.4.5 練習問題

食堂の担当者は、F 定食を 200 枚、M 定食を 300 枚、V 定食を 500 枚で収束させたいと考えています (他の想定も考えられます)。初期の各定食の枚数と行列 A を適当に定めなさい。条件として、つぎのようなことを考慮してください。

- 同じ定食の券が連続して配布するのは少なめにしてください。
- 各食券での次の日の食券の確率は、食券毎にできるだけ異なるの値で設定してください。

(各食券で、行列の縦の成分は異なるようにしてください。同じ確率だと、簡単に推測されてしまうため)

*4 「 t 」は、行ベクトル表記を列ベクトル表記に転置する記号です

- 製造能力から考え 1 つの定食に集中するのは避けるように作ってみてください.

答えは無数にあります. いろいろな行列 A で, 上の条件をある程度満たして, 収束させたい値に収束します.

4.4.6 練習問題

学生の要求により, メニューを増やすことにしました. M 定食を止め, P 定食 (豚肉料理) と C 定食 (鶏肉定食) を追加しました. そのマルコフ連鎖のシミュレーションを行いなさい. 人数は 1000 人とします.

- (1) 図 11 のような図を作成なさい.
- (2) 表 1 のような表を作成しなさい.
- (3) Excel の計算表を作成なさい (例題の表をコピーして使うと便利)
- (4) 食券の枚数の変化のグラフを作成しなさい.
- (5) 定常状態の人数の目標を定め, 定常状態が目標に近くなる, 行列 A を作成しなさい.